/*
 * Copyright (c) 2009-2010 Florian Nuecke
 * 
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the "Software"), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 * 
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 */

package djboom.utils {
	import flash.utils.Dictionary;

	/**
	 * Utility class for using weak references.
	 * 
	 * @author fnuecke
	 */
	public class WeakReference {
		/**
		 * Dictionary used to implement the weak referencing.
		 */
		private var dict:Dictionary;

		/**
		 * Creates a new weak reference to the given object.
		 * 
		 * @param object
		 *				the object for which to create a weak reference.
		 */
		public function WeakReference(object:Object) {
			dict = new Dictionary(true);
			dict[object] = true;
		}

		/**
		 * The element referenced to by this weak reference, or <code>null</code>
		 * if the referenced object has been garbage collected.
		 */
		public function get value():Object {
			for (var obj:Object in dict) {
				return obj;
			}
		
			return null;
		}

		public function dispose():void {
			dict = null;
		}
	}
}
